Exercices d'entrainement de Pspice

On se propose ici de se familiariser avec 3 analyses usuelles en électronique : OP, AC, TRAN. C'est en tant qu'électronicien qu'il faut raisonner : étude théorique du schéma, simulation, et interprétation quantitative des résultats. On donne 3 exemples pour commencer.

1. Un exemple de l'utilisation de .OP, à reproduire sur sa machine

Soit le schéma ci-dessous :



On donne R1 = 3,3 kΩ ; R2 = 6,8 kΩ ; R3 = 4,7 kΩ.

On désire connaitre le potentiel continu en s quand e_1 = 1 V et e_2 = 2V

a) Etude théorique :

Le circuit étant un système linéaire, on peut appliquer le théorème de superposition :
La tension en s = contribution de la source e_1 quand e_2 est inactive + contribution de la source e_2 quand e_1 est inactive. Soit en posant $R_{23} = R_2//R_3$ et $R_{13} = R_1//R_3$, on a :

$$s = e_1 \frac{R_{23}}{R_1 + R_{23}} + e_2 \frac{R_{13}}{R_2 + R_{13}} .$$

L'application numérique donne : $R_{23} \approx 2{,}78$ kΩ ; $R_{13} \approx 1{,}94$ kΩ et par suite :

$$s \approx 1 \frac{2{,}78}{3{,}3 + 2{,}78} + 2 \frac{1{,}94}{6{,}8 + 1{,}94} \approx 0{,}457 + 0{,}444 = 0{,}901.$$

b) Simulation :

On place des tensions continues sur e_1 et e_2.
On choisit la directive .OP pour cette analyse en continu.

Le fichier de sortie (.OUT) montre :

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(1)	1.0000	(2)	2.0000	(3)	.9009

Le nœud n°1 est la source d'entrée 1.
Le nœud n°2 est la source d'entrée 2.
Le nœud n°3 est la sortie s.

```
circuit sommateur à résistances
* fichier circuit_R1R2R3.cir
V1 1 0 DC=1
V2 2 0 DC=2
R1 1 3 3.3k
R2 2 3 6.8k
R3 3 0 4.7k
.OP
.end
```

c) Interprétation :

V(1) et V(2) ont bien les valeurs des sources de tensions.
On retrouve en V(3) : 0,9009 V, soit la valeur calculée, aux arrondis près.

• Le fichier de sortie donne d'autres renseignements, comme les courants continus débités par chaque source :
Le signe négatif s'explique par la convention utilisée par Pspice : quadripôle ou « égoïste » : tout ce que qui sort est négatif.

VOLTAGE SOURCE CURRENTS

NAME	CURRENT
V1	-3.004E-05
V2	-1.616E-04

• On peut vérifier le courant dans R_3. Par loi des nœuds :
30,04 µA + 161,6 µA = 191,64 µA, valeur que l'on retrouve par loi d'ohm en s : 0,9009/4,7k.

• Par curiosité, on peut également interpréter une autre information donnée dans le fichier de sortie :

```
TOTAL POWER DISSIPATION   3.53E-04  WATTS
```

Par loi de Joule sur chaque résistance :

$$
\begin{aligned}
& 3{,}3\ \text{k}\Omega \times (30{,}04\ \mu\text{A})^2 \\
+\ & 6{,}8\ \text{k}\Omega \times (161{,}6\ \mu\text{A})^2 \\
+\ & 4{,}7\ \text{k}\Omega \times (191{,}64\ \mu\text{A})^2 \\
\approx\ & 3\ \mu\text{W} + 177{,}6\ \mu\text{W} + 172{,}6\ \mu\text{W} = 353{,}2\ \mu\text{W}
\end{aligned}
$$

2. Un exemple de l'utilisation de .AC, à reproduire sur sa machine

Soit le schéma ci-dessous :



On donne R = 3,3 kΩ ; C = 10 nF.

On désire connaitre la réponse harmonique de $\frac{s(j\omega)}{e(j\omega)}$ et savoir dans quelle plage ce montage est assimilable à un intégrateur.

a) Etude théorique :

En régime harmonique, la variable est la pulsation, ou la fréquence. On manipule les impédances complexes.

Le pont diviseur de tension s'écrit $H(j\omega) = \frac{s(j\omega)}{e(j\omega)} = \frac{1/(jC\omega)}{R + 1/(jC\omega)} = \boxed{\frac{1}{1 + j\omega\tau}}$ avec $\tau = RC$

On peut tracer 20 log$|H(j\omega)|$ en s'aidant des asymptotes :
Pour $\omega \to 0$, $H(j\omega) \approx 1$, donc 20 log du module est un plateau à 0 dB.
Pour $\omega \to \infty$, $H(j\omega) \approx 1/j\omega\tau$, donc 20 log du module est une asymptote à -20 dB/décade.
L'intersection des asymptotes est à la pulsation $1/\tau$, c'est-à-dire à la fréquence $1/(2\pi\tau)$.

b) Simulation :

La constante de temps est $\tau = RC = 33\ \mu s$.
Pspice utilise la fréquence (variable exploitée en électronique) et non la pulsation (plus usuelle en automatique).
La réponse harmonique doit être choisie dans une gamme de fréquence intéressante pour notre circuit : de part et d'autre de $1/(2\pi\tau) = 4822$ Hz est judicieux. Par exemple de 100 Hz à 100 kHz.
Une résolution de 1000 points par décade est largement suffisante.

Le nœud n°1 est la source d'entrée.
Le nœud n°2 est la sortie u_C.

```
circuit R C
* fichier circuit_RC.cir

Ve 1 0 AC=1
R1 1 2 3.3k
C1 2 0 10n

.AC DEC 1000 100 100k
.probe
.end
```



c) Interprétation :

On retrouve le plateau à 0 dB en basse fréquence, et la pente à - 20 dB/décade (ou - 6dB/octave) en hautes fréquences.

Le curseur positionné à – 3 dB permet de vérifier la fréquence de cassure placée à 4822 Hz.

Trace Name	Y1
X Values	4.8195K
DB(V(2))	-2.9973

En effet, à la pulsation $1/\tau$, le complexe $\dfrac{1}{1+j\omega\tau}$ s'écrit $\dfrac{1}{1+j}$ dont le module vaut $1/\sqrt{2}$.

Le passage à – 3 dB est appelé la fréquence de coupure. Dans le cas d'un ordre 1, c'est la fréquence d'intersection des asymptotes placée à $1/(2\pi\tau)$.

Le circuit est assimilable à un intégrateur (dont la transmittance est $\dfrac{1}{j\omega\tau}$) pour des fréquences supérieures à $1/(2\pi\tau)$.

Remarque :

on peut également tracer la phase par P(V(2)) : on reconnait la courbe d'équation -artg $\omega\tau$, dont la traversée à -45 ° se situe à la pulsation $1/\tau$, c'est-à-dire la fréquence 4822 Hz.

3. Un exemple de l'utilisation de .TRAN, à reproduire sur sa machine



Soit le schéma ci-contre :
On désire connaitre la forme d'onde $u_R(t)$ quand e(t) est un échelon de tension d'amplitude E_0 = 3 V. Conditions initiales nulles.
R = 50 Ω, L = 1 mH.

a) Etude théorique :

Fléchons courant, tension et écrivons toutes les lois électriques indépendantes :

$e(t) = u_L(t) + u_R(t)$, $u_R(t) = R\, i(t)$ $u_L(t) = L\, di/dt$

Il vient : $L\, di/dt + R\, i(t) = e(t)$.



• On reconnait une équation différentielle linéaire, dont la variable est i(t), à coefficients constants, avec second membre. La résolution passe par 4 étapes : intégration de l'équation sans second membre, recherche d'une solution particulière, puis solution complète, et application de la condition initiale, ce qui aboutit à : $i(t) = E_0 /R[\, 1 - \exp(-t/\tau)\,]$ pour $t > 0$, avec $\tau = L/R$

On déduit : $\boxed{u_R(t) = E_0\, [\, 1 - \exp(-t/\tau)\,]}$.

• On peut également résoudre en passant par les transformées de Laplace :

On pose $E(p) = L\{e(t)\}$, $S(p) = L\{s(t)\}$, $I(p) = L\{i(t)\}$

$E(p) = U_L(p) + U_R(p)$ $U_R(p) = R\, I(p)$ $U_L(p) = L\, p\, I(p)$

Il vient $E(p) = (R + L\,p)\, I(p)$ d'où $I(p) = \dfrac{E(p)}{R+Lp}$ avec $E(p) = \dfrac{E_0}{p}$ soit $I(p) = \dfrac{E_0}{p(R+Lp)}$.

Par décomposition en éléments simples : $I(p) = \dfrac{E_0}{R}\left(\dfrac{1}{p} - \dfrac{1}{(R/L+p)}\right)$. En exploitant les tables de transformées de Laplace, on retrouve : $i(t) = E_0 /R[\ 1 - \exp(-t/\tau)\ ]$ pour $t > 0$, avec $\tau = L/R$

On déduit $\boxed{u_R(t) = E_0 [\ 1 - \exp(-t/\tau)\ ]}$.

b) Simulation :

La constante de temps est $\tau = L/R = 20$ µs.
Pour faire l'échelon, on choisit une source pulse dont le front montant est de durée très brève (1 ns), et dont la durée à l'état haut de 3 V doit être bien supérieure à 20 µs : on choisit 200 µs. La période de répétition est de 400 µs.
Le run doit montrer au moins une centaine de µs après le front, pour apprécier l'évolution de $u_R(t)$ jusqu'à la valeur finale. On choisit 400 µs, ce qui permet de visualiser également l'évolution de la sortie lors du front descendant.
On choisit de décaler l'instant de départ de l'échelon, pour une meilleure lisibilité d'une part, et pour vérifier qu'il n'y a pas d'influence d'éventuelles conditions initiales d'autre part. L'origine des temps est donc à 100 µs.

Le nœud n°1 est la source d'entrée.
Le nœud n°2 est la sortie $u_R(t)$.

```
circuit L R
* fichier circuit_LR.cir

Ve 1 0 pulse (0 3 100u 1n 1n 200u 400u)
L1 1 2 1m
R1 2 0 50

.tran 10n 400u 0 10n
.probe
.end
```



c) Interprétation :

La sortie commence à 0 V, valeur de la condition initiale.
L'évolution de $u_R(t)$ suit une loi en $[1 - \exp(-t/\tau)]$. Pour vérifier la valeur numérique de τ, on peut appliquer la méthode des 63 % : à $t = \tau$, le terme $[1 - \exp(-t/\tau)]$ vaut 0,63.

Il suffit de placer le curseur à 63 % de l'évolution de $u_R(t)$, soit 1,89 V.
On a bien le passage à 1,89 V de V(2) 20 µs après le front de V(1). On a ainsi vérifié la valeur de τ.

Trace Color	Trace Name	Y1	Y2	Y1 - Y2
	X Values	120.005u	100.000u	20.005u
CURSOR 2	V(1)	3.0000	2.4000u	3.0000
CURSOR 1	V(2)	1.8966	330.405p	1.8966

Remarque :
après la simulation, on peut demander le tracé de la courbe : 3*(1-exp(-(time-100e-6)/20e-6))
on est en parfaite coïncidence avec V(2), pour t dans [100 µs ; 300 µs],
ce qui confirme l'équation $E_0 [\ 1 - \exp(-t/\tau)\ ]$.

Appliquer la même démarche (théorie, simulation, interprétation chiffrée) pour les montages suivants :

Exo 1



Soit le schéma ci-contre.
On désire connaitre le potentiel continu en s quand e_1 = 1 V, e_2 = 2V, I = 1 mA. Calculer, simuler, interpréter.
On donne :
R1 = 3,3 kΩ ; R2 = 6,8 kΩ ; R3 = 4,7 kΩ ; R4 = 2,2 kΩ.

Exo 2

On désire connaitre la résistance équivalente entre les nœuds A et B.
Calculer, simuler, interpréter.



Exo 3



Soit le schéma ci-contre.
On désire savoir dans quelle plage ce montage est assimilable à un dérivateur. Calculer, simuler, interpréter.
On donne R = 4,7 kΩ ; C = 2,2 nF.

Exo 4



Soit le schéma ci-contre.
Calculer la forme d'onde s(t) quand e(t) est un échelon de tension d'amplitude E_0 = 3 V. Conditions initiales nulles.
On donne R1 = 3,3 kΩ ; R2 = 6,8 kΩ ; R3 = 4,7 kΩ ; C = 100 µF.
Simuler, interpréter.

Exo 5



Soit le schéma ci-contre.
La source $e_1(t)$ est une tension continue de 2 V.
La source $e_2(t)$ est une sinusoïde d'amplitude 1 V, de fréquence 30 kHz.
On veut connaitre s(t) en régime établi : Calculer, simuler, interpréter.
On donne R = 4,7 kΩ ; C = 2,2 nF ; R1 = 4,7 kΩ

Exo 6

Soit le circuit RLC ci-dessous, dont la transmittance S(p)/E(p) est :

$$\frac{1}{\frac{p^2}{\omega_0^2}+\frac{2\,z}{\omega_0}p+1}$$



On donne R = 500 Ω ; L = 200 mH ; C = 1,2665 nF.

• par identification, donner :
f_0, (= $\omega_0/2\pi$) sa fréquence propre sans amortissement,
z, son coefficient d'amortissement réduit,
f_R, sa fréquence de résonance.

• On applique en e(t) un signal périodique dont la forme est :



temps de montée : tm = 30 µs
temps de descente : td = 30 µs
temps à l'état haut : th = 20 µs
temps à l'état bas : tb = 20 µs
niveau haut = 1 V
niveau bas = -1 V.

On veut connaitre s(t) en régime établi.
Calculer, simuler, interpréter..